Kinematics-Chapter-2.nb

Written in *Mathematica* 12.2.0.0 Vince Cronin
Begun February 13, 2021, with some material from earlier documents; revised Feb 28.

2.1 What is the shape and size of Earth?

2.1.1 What is the mean (average) radius of Earth?

The mean radius of Earth is 6371.01 ± 0.2 km (Yoder, 1995).

2.1.2 What is Earth's ellipsoid of rotation (reference ellipsoid), and how does the radius of the ellipsoid vary at the poles and equator?

https://en.wikipedia.org/wiki/Reference_ellipsoid

The following values are from Yoder (1995, p. 8)

- mean equatorial radius (Geodetic Reference System, 1980): $a = 6378.137$ km
- polar radius (GRS, 1980): $b = 6356.752$ km
- flattening $f = (a - b) / a$ (GRS, 1980): $f = 1 / 298.257222$

The difference between the mean equatorial radius and the polar radius is approximately 21.385 km, or about 0.336% of the mean radius of Earth.

2.1.3 What is the geoid?

https://en.wikipedia.org/wiki/Geoid

The geoid is an equipotential surface associated only with gravity and the rotation of Earth (*i.e.*, not due to tides, wind, or other external forcings). The geoid is a smooth but irregular surface whose irregularity is due to the uneven distribution of mass in Earth's interior. At any point on the geoid surface, the gravitational force acts perpendicular to the surface. Positive gravity anomalies that indicate excess mass result in the geoid surface being above the reference ellipsoid; negative anomalies (mass deficit) are indicated by the geoid surface located below the reference ellipsoid. The geoid is determined using gravity observations from orbital satellites (Cazenave, 1995).

The geoid surface varies from 106 m below the reference ellipsoid (SW of Sri Lanka) to 85 m above (Iceland) the reference ellipsoid in Iceland, so the variation in geoid surface elevation spans about 191 m, or about 0.9% of the variation in radii of the reference ellipsoid.

2.1.4 What is the range of topographic/bathymetric elevations on Earth?

The highest point relative to mean sea level on Earth is the summit of Chogolungma (Mt. Everest; 8848.86 m) in the Himalaya. The lowest point is the Challenger Deep in the Marianas Trench, -10,920 ± 10 m.

The variation in topographic elevation on Earth's surface spans about 19,769 m.

2.1.5 What is the shape of Earth that we assume for the purpose of plate - kinematic modeling?

A spherical Earth is assumed for kinematic modeling in this primer, in the interest of modeling simplicity.

Aside from making the mathematics of plate kinematics simpler for most people to work with, the assumption of a spherical Earth is spatially reasonable. The largest variation from a sphere among the non-spherical shapes attributed to Earth is seen in it's reference ellipsoid (a difference of 21.385 km between polar and equatorial radii), compared with the variations in topography (19.769 km between Chogolungma and the Challenger Deep) or geoid height (0.191 km between Iceland and the depression SW of Sri Lanka).

Imagine that we drew a perfect circle with a 100 mm radius, using a one-ought (0) drafting pen whose line is 0.35 mm wide. Such a circle would just fit on a typical letter-sized piece of paper. Let this circle represent a slice through the center of a spherical Earth and through its poles. At that scale (1:63,710,100) the difference between the polar and equatorial radii of the reference ellipsoid would be about 0.336 mm. That is, the total divergence from the circle would be contained within the width of the one-ought line that defines the circle.

Earth is as spherical as a billiard ball. The World Pool-Billiard Association specifies "All balls must ... measure 2 ¼ (±.005) inches [5.715 cm (± .127 mm)] in diameter" (https://wpapool.com/equipment-specifications/). Earth's reference ellipsoid would be within the tolerance of a billiard ball if Earth's mean diameter was 2.25 inches or 5.715 cm.

2.2 Getting familiar with vectors

2.2.1 What is a vector?

A **vector** is a list of numbers. If the list has three **elements** or **components** (each of which is a real number), we say that the vector has three **dimensions**. It suits our purposes in the study of plate kinematics to work with 3-dimensional vectors.

The quantitative description of any vector requires a **coordinate system** so that characteristics like length and direction can be measured. The coordinate systems that we will use in our study of plate kinematics will be right orthogonal coordinate systems, also known as Cartesian coordinate systems in honor of René Descarte. The three coordinate axes are perpendicular to each other.

We can conceptualize a vector as something like a line segment connecting two distinct points that do not share the same location. Let's call those two points *P* and *Q* for convenience. Line segments have a length or **magnitude** that is the distance between the two points. Vectors are characterized by both the distance between points and the **direction** from one point to the other as, for example, from *P* to *Q* or from *Q* to *P*.

For the sake of this initial explanation, let's represent the vector from *P* to *Q* with the symbol $\overrightarrow{PQ}$ and agree that $\overrightarrow{PQ}$ has a definite magnitude (the distance between *P* and *Q*) and direction (from *P* to *Q*). Plainly, the distance from *P* to *Q* is the same as the distance from *Q* to *P*, but the direction is opposite. Vector $\overrightarrow{PQ}$ is the **inverse** of vector $\overrightarrow{QP}$; that is, $\overrightarrow{PQ} = -\overrightarrow{QP}$.

Now imagine two other points, *S* and *T*, and the associated vector $\overrightarrow{ST}$. If $\overrightarrow{PQ}$ has the same magnitude and direction as $\overrightarrow{ST}$, then we say that vector $\overrightarrow{PQ}$ is equivalent to vector $\overrightarrow{ST}$; that is, $\overrightarrow{PQ} = \overrightarrow{ST}$.

One way of making sense of this equivalency is by insisting that the origin of a coordinate system used to characterize the magnitude and direction of a vector be the same as the "origin point" of the vector — for example, point *P* of the vector $\overrightarrow{PQ}$. With *P* at the origin (P = {0, 0, 0}), the three coordinates of the "destination point" *Q* within the specified Cartesian coordinate system (Q = $\{x_Q, y_Q, z_Q\}$) supply the three elements of numerical description of the vector. We sometimes refer to the coordinate system axes as the *X*, *Y*, and *Z* axes, respectively, or as the 1, 2, and 3 axes. Transitioning to the way that we will represent vectors elsewhere in this text, let us re-name vector $\overrightarrow{PQ}$ as vector *a* that extends from the origin of a Cartesian coordinate system to an endpoint whose coordinates are $\{a_1, a_2, a_3\} = \{x_Q, y_Q, z_Q\}$. That is, vector *a* extends from the point {0, 0, 0} to the point $\{a_1, a_2, a_3\}$, so vector $a = \{a_1, a_2, a_3\}$.

Finally, we note that a **vector** is a first-rank tensor and a **scalar** is a zero-rank tensor (that is, a scalar has a magnitude but no direction).

In *Mathematica*, lists of numbers or variables are usually contained within curly brackets, so in defining a vector a we might write the following code.

```
vectorA = {2,5,3}
```

The traditional matrix form of a vector depicts the elements of the vector in a column, with the first-row element corresponding to the X coordinate of the vector, the second-row element is the Y coordinate of the vector, and the third/bottom-row element is the Z coordinate.

vectorAMatrixForm = $\begin{pmatrix} 2 \\ 5 \\ 3 \end{pmatrix}$

2.2.2 What is a location vector?

As we use the term in plate kinematics, a **location vector** extends from the origin of the coordinate system at the center of Earth to a specified point on Earth's surface. Such location vectors are often assigned a length of one Earth radius.



Figure 2.1. The Cartesian geographic coordinate system. The Z_{GEOG} axis passes through the geographic North Pole, the XGEOG axis contains the intersection of the Prime Meridian and the Equator, and the Y_{GEOG} axis is normal to the Prime Meridian. Positive longitude is measured east of the Prime Meridian; positive latitude is measured north of the Equator. The Cartesian coordinates of a point Q, whose longitude and latitude are known, are $\{x_Q, y_Q, z_Q\}$. From Cronin (1991).

2.2.3 How do we determine the location vector to a point whose geographic coordinates (latitude and longitude) are given?

Geographic coordinates of latitude and longitude are typically presented in one of three ways: as decimal degrees (e.g., 35.2546°), as degrees and decimal minutes (e.g., 35°15.276'), or as degrees, minutes, and decimal seconds (e.g., 35°15'16.56"). There are 60 minutes per degree of arc and 60 seconds per minute of arc. Examples of the conversion to decimal degrees follow.

EXAMPLE. Convert the given geographic coordinates to decimal geographic coordinates , where ' indicates minutes and " indicates seconds of arc.

- 35°15.276': $35° + \left(\frac{15.276'}{60'/°}\right) = 35.2546°$
- 35°15'16.56": $35° + \left(\frac{15'}{60'/°}\right) + \left(\frac{16.56''}{3600''/°}\right) = 35.2546°$

Given the decimal latitude and decimal longitude of a point Q, the coordinates in a Cartesian geographic coordinate system are

$x_Q = \cos(\text{latitude}_Q)\cos(\text{longitude}_Q)$

$y_Q = \cos(\text{latitude}_Q)\ \sin(\text{longitude}_Q)$

$z_Q = \sin(\text{latitude}_Q)$

The corresponding location vector to point Q is

$Q = \{x_Q, y_Q, z_Q\}$

In *Mathematica*, the built-in trigonometric functions `Cos[]` and `Sin[]` act on an angle expressed in radian measure. If the input to these functions is typically expressed in degrees, we add the built-in function `Degree` to the argument within the square brackets to convert from degrees to radians as follows.

```
xQ = Cos[latitudeQ Degree] Cos[longitudeQ Degree];
yQ = Cos[latitudeQ Degree] Sin[longitudeQ Degree];
zQ = Sin[latitudeQ Degree];

vectorQ = {xQ, yQ, zQ}
```

The user-defined function `makeLocationVector[]` can be inserted in a Mathematica notebook to perform the conversion from geographic to Cartesian geographic coordinates. The arguments supplied to this function include the latitude and longitude of the point, expressed in decimal degrees. Those input arguments are modified by the built-in function `Degree` to convert from decimal degrees to radians.

```
makeLocationVector[lat_, long_]:={Cos[lat Degree] Cos[long Degree],
Cos[lat Degree] Sin[long Degree], Sin[lat Degree]};
```

Example. We could define a location vector for a point P whose geographic coordinates are 35.248°N latitude and 119.958°W longitude using the following code in association with the user-defined function `makeLocationVector[]`.

First, we establish the user-defined function.

```
makeLocationVector[lat_, long_]:={Cos[lat Degree] Cos[long Degree],
Cos[lat Degree] Sin[long Degree], Sin[lat Degree]};
```

Second, we define the input data. Notice that west longitudes and south latitudes are negative numbers.

```
latP = 35.248;
longP = -119.958;
```

Third, we execute the computation to define the location vector to point P.

```
vectorP = makeLocationVector[latP, longP];
```

Finally, if we want to see the numerical result of the computation, we add the following line that utilizies the built-in Mathematica function `N`, without a semicolon at the end.

```
N[vectorP]
```

Exercise 2.1 (HW-01) Use Google Earth to find the latitude and longitude of some place that is of interest to you, using decimal degrees. Create a *Mathematica* notebook with headers for its title, intro-

> duction, input data, computation, output, and references (if any). Save the notebook using your last name as the first part of the notebook name, then the homework number. For example, “Jackson-HW01”. Add text to the notebook providing your name, the date you began and completed the notebook, a statement of the problem, and relevant explanations of the variables and code. Finally, add input lines of executable code. The purpose of this notebook is to compute the unit location vector to the point that is of interest to you.

2.2.4 What is a zero vector?

If all of the elements of a vector are zeros, that vector is called a **zero vector**.

2.2.5 How do we determine the length of a vector?

Imagine we have a vector a such that its coordinate along the x (or 1) coordinate axis is the element a_1, its coordinate along the y (or 2) axis is a_2, and its coordinate along the z (or 3) axis is a_3.

$a = \{a_1, a_2, a_3\}$

We can use the Pythagorean Theorem to determine the length of that vector, which we indicate using double bracket bars.

$||a|| = \sqrt{(a_1^2 + a_2^2 + a_3^2)}$

The length of a vector is also called its magnitude. If the vector represents a velocity, the length of that vector is called the speed.

EXAMPLE. Let a = {5, 4, 2}, then the length of a is

$||a|| = \sqrt{5^2 + 4^2 + 2^2} = \sqrt{45}$

In *Mathematica*, the built-in function `Norm` provides the length or magnitude of a vector.

```
length_a = Norm[a];
```

2.2.6 What is a unit vector?

A vector whose length is 1 is called a **unit vector**.

2.2.7 How do we determine the unit vector that is parallel to a non - zero vector?

The unit vector is determined by dividing each element of the vector by the vector’s length. Imagine we have a vector a such that

$a = \{a_1, a_2, a_3\}$.

The length of vector a is

$\|a\| = \sqrt{(a_1{}^2 + a_2{}^2 + a_3{}^2)}$.

The unit vector that coincides with vector a is designated by $\hat{a}$ (called "a hat") and is given by

$$\hat{a} = \frac{a}{\|a\|} = \left\{\frac{a_1}{\|a\|}, \frac{a_2}{\|a\|}, \frac{a_3}{\|a\|}\right\}$$

EXAMPLE. Let $a = \{5, 4, 2\}$. The length of a is

$\|a\| = \sqrt{5^2 + 4^2 + 2^2} = \sqrt{45}$

The unit vector that is parallel with (or coincides with) vector a is

$$\hat{a} = \left\{\frac{5}{\sqrt{45}}, \frac{4}{\sqrt{45}}, \frac{2}{\sqrt{45}}\right\} \simeq \{0.7454, 0.5963, 0.2981\}$$

In *Mathematica*, the built-in function **`Normalize`** provides the unit vector that coincides with the specified vector.

```
unitVectorA = Normalize[vectorA];
```

2.2.8 What is a dot product?

Given two non-zero vectors, a and b where

$a = \{a_1, a_2, a_3\}$.

and

$b = \{b_1, b_2, b_3\}$,

the **dot product** (also known as the **inner product** or the **scalar product**) can be defined in several equivalent ways. These include

$a \cdot b = a^T b$

$a \cdot b = [(a_1\, b_1) + (a_2\, b_2) + (a_3\, b_3)]$, and

$a \cdot b = \|a\|\, \|b\| \cos(\theta)$

where θ us the angle between the two vectors.

In *Mathematica*, the built-in function **`Dot[]`** acts on the two vectors listed between the square brackets and returns the dot product of the two vectors.

```
result1 = Dot[a,b];
```

Another way of obtaining the dot product in *Mathematica* is to put a period between the vectors.

```
result2 = a.b
```

While there is a "middle dot" symbol available in the *Mathematica* typesetting menu, this is not recognized as a functional character.

2.2.9 How do we determine the angle between two non - zero

vectors?

One of the definitions of the dot product provides a way to determine the angle θ between two non-zero vectors.

$a \cdot b = \|a\| \, \|b\| \cos(\theta)$

Rearranging the equation to isolate the angle θ, we have

$\theta = \cos^{-1}\left(\frac{a \cdot b}{\|a\| \, \|b\|}\right).$

If both a and b are unit vectors, the angle between $\hat{a}$ and $\hat{b}$ is given by

$\theta = \cos^{-1}(\hat{a} \cdot \hat{b}).$

In *Mathematica*, the built-in function `VectorAngle[]` acts on the two vectors listed between the square brackets and returns the angle between two vectors, returning the answer in radian measure. To convert this to degrees, the result can be multiplied by $(180/\pi)$

```
thetaRadian = VectorAngle[a,b]
thetaDegrees = VectorAngle[a,b] (180/π)
```

2.2.10 How do we convert a location vector to geographic coordinates (latitude and longitude)?

Toward the end of some kinematic computations, we will need to convert the Cartesian geographic coordinates of a location vector into geographic coordinates of latitude and longitude.

Let’s define a location vector a that we would like to convert to geographic coordinates.

$a = \{a_1, a_2, a_3\}$

The easiest part of the problem is to find the latitude, because

latitude $= \sin^{-1}(a_3) = \arcsin(a_3)$.

Imagine that we start with a point A whose geographic coordinates are latitude 35°N and longitude 125°E, and we want to know its coordinates in the Cartesian geographic coordinate system. In *Mathematica*, we might define those geographic coordinates as follows.

```
latA=35;
longA=125;
```

Based on the information in section 2.2.3, we define the function `makeLocationVector[]` as

```
makeLocationVector[lat_, long_]:={Cos[lat Degree] Cos[long Degree],
Cos[lat Degree] Sin[long Degree], Sin[lat Degree]};
```

We proceed to find the location vector `locVectA` as

```
locVectA=makeLocationVector[latA, latB];
```

The numerical approximation of the location vector **locVectA**, expressed to six places past the decimal

point, is
{-0.469846, 0.671010, 0.573576}

Example, part 1. The latitude (`latitudeA`) of the location vector (`locVectA`) is given by
`latitudeA=ArcSin[locVectA[[3]]] (180/π)`
where the third component of vector `locVectA` is designated by `locVectA[[3]]`, and the answer is expressed in degrees. We can use a hand calculator to verify that $\sin^{-1}(0.573576) \simeq 0.610865$ radian, and 0.610865 radian * (180°/π radian) $\simeq$ 35°.

If components a_1 and a_2 are both zeros and a_3 is a positive 1, the point is the north pole, whose latitude is 90° and longitude is undefined. If a_3 is –1 and the other components are zeros, the point is the south pole whose latitude is –90° and longitude is undefined.

Let's define a vector b as the projection of vector a onto the X-Y plane of the Cartesian geographic coordinate system; that is, onto the plane of Earth's Equator (also known as the equatorial plane).

$b = \{a_1, a_2, 0\}$

The length or magnitude of vector b , $\llbracket b \rrbracket$, is

$\llbracket b \rrbracket = \sqrt{(a_1^2 + a_2^2)}$

The X axis of the Cartesian geographic coordinate system passes through the intersection of the Prime Meridian and the Equator, and is the datum for determining the longitude along the Equator. We use the word "datum" in the sense of the point or line from which the distance (or angular distance) to other points or lines are measured. So the "datum" on an old-fashioned mechanical clock is the vector that points from the center of the clock face toward the "12." Let's define vector c as the location vector to the intersection of the Equator and the Prime Meridian, along the positive X axis of the Cartesian geographic coordinate system.

$c = \{1, 0, 0\}$.

Now, let's define vector θ as the angle between vectors b and c, based on information from section 2.2.9.

$\theta = \cos^{-1}\left(\frac{b \cdot c}{\|b\| \; \|c\|}\right)$

Because $b = \{a_1, a_2, 0\}$ and $c = \{1, 0, 0\}$, the dot product $(b \cdot c) = [(a_1 * 1) + (a_2 * 0) + (0 * 0)] = a_1$.

Noting that vector c is a unit vector (that is, $\llbracket c \rrbracket = 1$), we can simplify the computation of θ as follows:

$$\theta = \cos^{-1}\left(\frac{a_1}{\sqrt{(a_1^2 + a_2^2)}}\right)$$

If the value of vector component $a_2$is a positive value or zero (that is, if $a_2 \geq 0$),

longitude = θ

or if the value of vector component $a_2$is a negative value (that is, if $a_2 < 0$),

longitude = $-\theta$.

In *Mathematica*, the user-defined module `findGeogCoord[]` acts on a unit location vector in the Cartesian geographic coordinate system and yields the corresponding geographic coordinates (latitude and longitude).

```
findGeogCoord[vect_] := Module[{lat, long, a, b, θ}, a = ArcSin[vect[[3]]];
   θ = If[((Abs[vect[[1]]] < (1 × 10^-14)) && (Abs[vect[[2]]] < (1 × 10^-14))),
     0, ArcCos[vect[[1]] / Sqrt[(vect[[1]]^2 + vect[[2]]^2)]]];
   b = If[(vect[[2]] < 0), (-θ), (θ)];
   lat = a (180 / π);
   long = If[((Abs[vect[[1]]] < (1 × 10^-14)) && (Abs[vect[[2]]] < (1 × 10^-14))),
     0, (b (180 / π))];
   {lat, long}];
```

In the module above, we presume that any value that is smaller than 1×10^{-14} is effectively equal to zero.

Example, part 2. The longitude (`longitudeA`) of the location vector (`locVectA`) can be computed in a “longhand” manner as follows.
If `locVectA[[1]]` and `locVectA[[2]]` are both zero, the longitude is undefined. (That is, if the first and second components of the vector **locVectA** both equal zero, the longitude is undefined.) If `locVectA={0,0,1}`, the point is the North Pole (latitude 90°N, undefined longitude), and if `locVectA={0,0,-1}`, the point is the South Pole (Latitude -90° or 90°S, undefined longitude).
If `locVectA[[1]]` and `locVectA[[2]]` are *not* both zero, then the angle θ between the Prime Meridian and the projection of `locVectA` onto the Equatorial plane is given by
$\boldsymbol{\theta} = \text{ArcCos}\left[\textbf{locVectA[[1]]} \big/ \sqrt{(\textbf{locVectA[[1]]}^2 + \textbf{locVectA[[2]]}^2)}\right]$
If $x \geq 0$, `longitudeA=θ`; if $x < 0$, `longitudeA = -θ`.

It is considerably easier to solve for both the latitude and longitude associated with the given unit location vector `locVectorA` by applying the user-defined module `findGeogCoord[]` as follows.
`geogCoords=findGeogCoord[locVectorA];`

The full process of converting from geographic coordinates to Cartesian geographic coordinates and back again is illustrated in the following code.

```
latA = 35;
longA = 125;

makeLocationVector[lat_, long_] := {Cos[lat Degree] Cos[long Degree],
   Cos[lat Degree] Sin[long Degree], Sin[lat Degree]};
```

```
findGeogCoord[vect_] := Module[{lat, long, a, b, θ}, a = ArcSin[vect[[3]]];
   θ = If[((Abs[vect[[1]]] < (1 × 10^-14)) && (Abs[vect[[2]]] < (1 × 10^-14))),
     0, ArcCos[vect[[1]] / Sqrt[(vect[[1]]^2 + vect[[2]]^2)]]];
   b = If[(vect[[2]] < 0), (-θ), (θ)];
   lat = a (180 / π);
   long = If[((Abs[vect[[1]]] < (1 × 10^-14)) && (Abs[vect[[2]]] < (1 × 10^-14))),
     0, (b (180 / π))];
   {lat, long}];

locVectA = makeLocationVector[latA, longA];
N[locVectA]

geogCoords = findGeogCoord[locVectA];
N[geogCoords]
```

The result will be that the input data (**latA, longA**) will have the same values as the output data (**geogCoords**).

2.2.11 What is a vector cross product?

Imagine that we have two non-zero and non-colinear vectors, a and b.

$a = \{a_1, a_2, a_3\}$ and $b = \{b_1, b_2, b_3\}$.

The vector **cross product** $a \times b$ (stated as "a cross b") yields a third vector, which is perpendicular to both a and b.

$a \times b = \{(a_2\, b_3 - a_3\, b_2), (a_3\, b_1 - a_1\, b_3), (a_1\, b_2 - a_2\, b_1)\}$.

Let c represent the vector result of $a \times b$. Visualize vectors a and b on the plane that they share, where the angle measured in a positive (anti-clockwise) direction from a to b is less than 180° ($<\pi$ radian). Vector c extends perpendicular to that plane (normal to the plane) in the direction the extended thumb on your right hand points when your index finger curls from a to b (see Fig. 2.2).



Figure 2.2. Visualization of the geometry of a vector cross product. The result of the vector cross product $a \times b$ is vector c, which is oriented perpendicular to the a-b plane. The direction of c is given by the right-hand rule: the thumb of the right hand points in the direction of c, where vector b is an anticlockwise rotation of <180° from a around their common origin.

EXAMPLE. Given vectors $a = \{5, 4, 2\}$ and $b = \{3, 1, 6\}$. The result of the cross product $a \times b$ is vector c.

$$c = a \times b = \{((4*6)-(2*1)), ((2*3)-(5*6)), ((5*1)-(4*3))\}$$

$$c \simeq \{22, -24, -7\}$$

If now we consider the vector result of $b \times a$,

$$b \times a = \{(b_2 a_3 - b_3 a_2), (b_3 a_1 - b_1 a_3), (b_1 a_2 - b_2 a_1)\}.$$

we find that the result is another vector that is perpendicular to the a-b plane, with the same length (magnitude) as c but pointing in the opposite direction.

EXAMPLE. Given vectors $a = \{5, 4, 2\}$ and $b = \{3, 1, 6\}$. The result of the cross product $b \times a$ is vector d.

$$d = b \times a = \{((1*2)-(6*4)), ((6*5)-(3*2)), ((3*4)-(1*5))\}$$

$$d \simeq \{-22, 24, 7\}$$

In *Mathematica*, the built-in function `Cross[]` acts on the two vectors listed between the square brackets and returns the cross product of the two vectors.

```
result1=Cross[a,b]
```

Another way of obtaining the cross product in *Mathematica* is to select the “cross” symbol from the Basic Math Assistant pallette and insert it between the vector symbols.

```
result2=a×b
```

Take care not to select the “times” symbol from the Basic Math Assistant pallette if you want the cross-product function. The Basic Math Assistant pallette is accessible from the Pallettes drop-down menu. The “cross” symbol looks like this (×) and the “times” symbol looks like this (×). If you hover the cursor over the symbol in the Basic Math Assistant pallette, a box pops up and indicates the name of the symbol.

2.3 Beginning to use vectors for computation on a spherical Earth

2.3.1 What are great circles and small circles?

The intersection of a sphere with a plane that passes through the center of the sphere is called a **great circle**. Earth's equator is a great circle.

The prime meridian is a semi-circular arc along the great circle that includes the spin axis (north and south poles) of Earth.

If a plane intersects a sphere but does not pass through the center of the sphere, the intersection of the sphere and that plane is a **small circle** because its radius is smaller than that of the sphere. With the single exception of the equator, the parallels of latitude in the geographic coordinate system are all small circles.

Ignoring the effects of topographic and bathymetric obstacles, the shortest distance between two points on a spherical Earth is along a great circle containing those two points.

2.3.2 What does it mean to say that a point on Earth's surface is the *antipode* of another point?

The antipode of a point on Earth's surface is locate directly opposite that point on the other side of Earth. (The word "antipode" sounds like "Auntie pode" where "pode" rhymes with "code.") So a vector from a point on Earth's surface, extending downward through Earth's center and all the way to the other side, would point toward that point's antipode. Two points are antipodal if they are located along a great circle 180° (π radians) away from each other. The location vectors to two antipodal points are colinear, but pointed in opposite directions.

2.3.3 Given a spherical Earth of radius 6371.01 km, what is the circumferential distance (in km) corresponding to an angular distance of 1° of arc along a great circle?

The circumference of an assumed-spherical Earth of radius (r) 6371.01 km is as follows.

circumference = $2\,\pi\,r$ = 40,030.24 km

circumferential distance corresponding to 1° of arc = $\frac{40{,}030.24\text{ km}}{360\,°}$ = 111.20 km/°

circumferential distance corresponding to 1 radian of arc = $\frac{40{,}030.24\text{ km}}{2\,\pi}$ = 6371.01 km/rad

2.3.4 How do we determine the angular and great-circle (circumferential) distance between two points on Earth's surface?

Given two non-colinear unit location vectors, $\hat{a}$ and $\hat{b}$, the angular distance θ between $\hat{a}$ and $\hat{b}$ is given by
$\theta = \cos^{-1}(\hat{a} \cdot \hat{b})$
as described in sections 2.2.8 and 2.2.9.

In *Mathematica*, the angular distance between two unit location vectors *a* and *b* can be found using the built-in functions `VectorAngle[]` and π as follows:

```
thetaRadians=VectorAngle[a,b]
thetaDegrees=VectorAngle[a,b] (180/π)
```

Knowing the angular distance θ between $\hat{a}$ and $\hat{b}$, the great-circle arc distance (circumferential distance) in km is given by
distance = θ rad * 6371.01 km/rad
if the angular distance is expressed in radians, or
distance = θ ° * 111.20 km/°
if the angular distance is expressed in degrees.

In *Mathematica*, the circumferential or great-circle distance between two unit location vectors *a* and *b* can be found using the built-in functions `VectorAngle{]` and π as follows:

```
circumfDist1=VectorAngle[a,b]*6371.01
circumfDist2=VectorAngle[a,b]*(180/π)*111.20
```

where the mean radius of Earth is 6371.01 km and the circumferential or great-circle distance between two points on Earth's surface is 111.20 km/° of angular distance.

Exercise 2.2 (HW-02) In the early Miocene ~23 million years ago, a volcano erupted in California. Sometime later in the Miocene, the San Andreas fault propagated through the volcanic field, and separated it into what is now the Pinnacles National Monument (36°29'13"N, 121°10'01"W) on the west side of the fault and the Neenach volcanic field (34°44'24"N, 118°37'24"W) on the east side (Matthews, 1973).
Write a *Mathematica* notebook that analyzes the input data to complete the following tasks.
(a) Convert the given geographic coordinates to decimal geographic coordinates (see section 2.2.3).
(b) Convert the decimal geographic coordinates to unit location vectors, recalling that south latitudes and west longitudes are negative numbers.
(c) Determine the angular distance between the Pinnacles and Neenach (see section 2.2.9).
(d) Find the circumferential distance between the Pinnacles and Neenach, assuming that Earth is a sphere of radius 6,371.01 km (see section 2.3.4).
(e) What broad geological statement(s) can you make about the rate and magnitude of displacement along the San Andreas fault, based only on the information provided and your calculations?

2.4 Working with planes and great circles through a spherical Earth

2.4.1 How do we determine the vector normal to a plane defined by two non-colinear non-zero vectors?

When we refer to a vector *a* that is *normal* to a plane, we mean that vector *a* is *perpendicular* to a plane. Indeed, both vector *a* and its inverse, vector $-a$, are perpendicular to the plane, although they are pointed in opposite directions. (If vector a = $\{a_1, a_2, a_3\}$, then $-a = -1*a = \{-a_1, -a_2, -a_3\}$.) If we require that the plane include the center of a sphere — the origin of the coordinate system — the intersection of the sphere and the plane is a *great circle*, as we have previously noted. The *pole* of the great circle is the place on the sphere for which the unit vector normal to the plane, vector $\hat{a}$, is the location vector. The *antipole* corresponds to the unit location vector $-\hat{a}$.

Imagine that we have two non-zero and non-colinear vectors, *b* and *c*. Because *b* and *c* have the same origin (that is, they extend from the same point at the origin of a coordinate system), there is one and only one plane that includes both *b* and *c*. Vectors *b* and *c* are coplanar.

If *b* and *c* are location vectors, the respective two points on the sphere are located along a great circle.

A vector *a* that is normal to the plane defined by the non-zero and non-colinear vectors *b* and *c* is defined by the vector cross product

$a = b \times c$

The other vector normal to that plane, $-a$, is defined by the vector cross product

$-a = c \times b$

Vectors *a* and $-a$ are colinear with each other, point in opposite directions, and are both normal to the plane defined by vectors *b* and *c*.

In *Mathematica*, the vectors normal to non-zero, non-colinear vectors *b* and *c* can be found using the built-in function `Cross[]` as follows:

```
normalVectA=Cross[b,c]
normalVectMinusA=Cross[c,b]
```

where

```
normalVectMinusA=(-1)*normalVectA
```

2.4.2 How do we determine the (conjugate) dihedral angles between two planes that pass through the origin of the coordinate system?

The angle between the vector normal to one plane (call it “plane 1”) and the vector normal to a different plane (call it “plane 2”) is the same as the dihedral angle between plane 1 and plane 2. If we call the dihedral angle θ, the conjugate dihedral angle is equal to $180° - \theta$ in degrees or $\pi - \theta$ rad in radians.

EXAMPLE. Given four unit location vectors $(\hat{a}, \hat{b}, \hat{c}, \text{ and } \hat{d})$ such that no two of these vectors are colinear and no three of these vectors are coplanar. What are the conjugate dihedral angles between the plane defined by vectors $\hat{a}$ and $\hat{b}$ (plane 1) and the plane defined by vectors $\hat{c}$ and $\hat{d}$ (plane 2)?

Let's define vector m as the vector normal to plane 1. Based on what we learned in section 2.4.1,

$m = \hat{a} \times \hat{b}$.

Define vector n as the vector normal to plane 2, such that

$n = \hat{c} \times \hat{d}$.

One of the two conjugate dihedral angles between plane 1 and plane 2 is θ, where (based on sections 2.2.8 and 2.2.9)

$$\theta = \cos^{-1}\left(\frac{m \cdot n}{\|m\|\ \|n\|}\right)$$

If θ is expressed in degrees, the conjugate dihedral angle is $180° - \theta$ °; alternatively, if θ is expressed in radians, the conjugate dihedral angle is $\pi - \theta$ rad.

In *Mathematica*, the dihedral angle between plane 1 and plane 2, as described in the preceding example, could be found as follows

```
thetaDegrees=VectorAngle[Cross[a,b],Cross[c,d]] (180/π)
```

or

```
thetaRadians=VectorAngle[Cross[a,b],Cross[c,d]]
```

The conjugate dihedral angle would then be

```
conjugateDegrees=180-thetaDegrees
```

or

```
conjugateRadians = π - thetaRadians
```

Exercise 2.3 (HW-03) Imagine yourself standing at Neenach in southern California (34°44'24"N, 118°37'24"W), and you want to know the direction of the shortest straight-line path to Pinnacles in west-central California (36°29'13"N, 121°10'01"W). ***The azimuth of a bearing is measured in a clockwise direction relative to true north.*** In a general sort of way, you could start by looking toward true north (90°N, 0°E) and slowly rotating clockwise while keeping track of your total rotation angle until you are looking directly toward Pinnacles (assuming you could see that far and over the curvature of Earth). If you could perform that angular measurement accurately, that total rotation angle would be the azimuth from Neenach to Pinnacles.

Judging from the geographic coordinates of Neenach and Pinnacles, would you have to rotate clockwise *less than* or *more than* 180° from true north to be looking along the shortest great-circle path to Pinnacles? You might want to sketch a map of the situation to form your judgement.

Write a *Mathematica* notebook that analyzes the input data to complete the following tasks.

(a) Convert the given geographic coordinates of Neenach and Pinnacles to decimal geographic coordinates (see section 2.2.3).

(b) Convert the decimal geographic coordinates to unit location vectors (**locVectNeenach** and **locVectPinnacles**), recalling that south latitudes and west longitudes are negative numbers.

(c) Determine the dihedral angle between (1) the plane defined by the location vector to Neenach and the location vector to the North Pole, and (2) the plane defined by the location vector to Neenach

> and the location vector to Pinnacles (see sections 2.4.1 and 2.4.2). For the location vector to the North Pole, define **locVectNorthPole** = {0,0,1}.
>
> (d) The dihedral angle you just determined is related in some quantitative way to the azimuth of the great-circle path from Neenach to Pinnacles. Conjure-up some spatial reasoning to determine that azimuth, and write a mathematical expression to indicate the relationship between the dihedral angle you computed and the azimuth *in this case*. For example,
>
> - azimuth = dihedral angle
> - azimuth = 360° – (dihedral angle)
> - azimuth = 180° + (dihedral angle)
>
> *et cetera*.
>
> Note that we will write some code so that future determinations of azimuth will not rely on your powers of spatial reasoning.

2.4.3 How do we determine the azimuth from one point to another point on Earth's surface?

The process of determining the azimuth from one point to another point on Earth's surface requires the unit location of those two points and the North Pole. The geographic coordinates of the North Pole are latitude 90° with a longitude that is undefined, but generally taken to be 0°. The Cartesian geographic coordinates of the unit location vector to the North Pole are {0, 0, 1}.

Let's define the reference point as the point where an imaginary observer is located, from which the azimuth of another point is determined. We'll refer to that other point as the target point. The unit location vectors we need to work with will be called r (to the reference point), t (to the target point), and n (to the North Pole).

Vector a is defined such that $a = n \times r$. Vector a is a vector normal to the plane defined by the North Pole and the reference point.

Vector b is defined such that $b = t \times r$. Vector a is a vector normal to the plane defined by the target point and the reference point.

Angle $\theta 1$ is the angle (expressed in degrees) between the vectors a and b. This is the same as one of the two conjugate dihedral angles between the plane defined by vectors n and r, and the plane defined by vectors t and r. (Adding two conjugate angles together sums to 180°.)

Angle $\theta 2$ is the angle (expressed in degrees) between the vectors a and t. Angle $\theta 2$ helps us define the azimuth of the bearing from the reference point to the other point. Imagine that Earth is divided into two hemispheres bounded by the plane that contains Earth's spin axis and the reference point. We will specify that the north pole is "up" in this frame of reference, so one hemisphere would be to the right of the reference point and the other would be to the left.

If the other point is located along the same circle of longitude as the reference point, then $\theta 2$ will equal 90° or 180°, and the bearing from the reference point to the other point will either be 0° or 180°.

If the other point is located on the hemisphere to the right of the reference point, then $\theta 2$ will be less than 90°. In that case, the azimuth of the bearing from the reference point to the other point is equal to $\theta 1$. Bearings are expressed in degrees relative to north, where due north is 0°, due east is 90°, due south is 180°, and due west is 270°.

If the other point is located on the hemisphere to the left of the reference point, then *θ2* will be greater than 90°. In that case, the azimuth of the bearing from the reference point to the other point is equal to (360°- *θ1*).

Vector *c* is the vector normal to the plane defined by the reference point (vector *r*) and vector *a*, such that $c = r \times a$.

Angle *θ3* is the angle (expressed in degrees) between the vectors *c* and *t*. The vector *c* and the angle *θ3* are used when the target point is on the same meridian as the reference point, to differentiate between a bearing of 0° (north) or 180° (south) from the reference point to the target point. If *θ3* is less than 90°, the target point is north of the reference point. If ***θ*3** is equal to 90°, the target point is either coincident with the reference point or 180° from the reference point, in which case the bearing from the reference point to the target point is undefined. For simplicity, when ***θ*3**=90°, we set the bearing to 0°.

In *Mathematica*, the user-defined module **findAzimuthVectInput[]** acts on the unit location vectors of a reference point and a target point in the Cartesian geographic coordinate system, yielding the azimuth (in degrees) from the reference point to the target point.

```
findAzimuthVectInput[refPtVect_, otherPtVect_] :=
  Module[{northPole, vectorA, vectorB, θ1, θ2, vectorC, θ3, azimuth},
   northPole = {0, 0, 1};
   vectorA = Cross[northPole, refPtVect];
   vectorB = Cross[otherPtVect, refPtVect];
   θ1 = VectorAngle[vectorA, vectorB] (180 / π);
   θ2 = VectorAngle[vectorA, otherPtVect] (180 / π);
   vectorC = Cross[refPtVect, vectorA];
   θ3 = VectorAngle[vectorC, otherPtVect] (180 / π);
   azimuth = If[((θ2 == 90) ∨ (θ2 == 180)),
     If[(θ3 <= 90), 0, 180], If[(θ2 > 90), (360 - θ1), θ1]];
   azimuth ];
```

Similarly, the user-defined module **findAzimuthGeogInput[]** acts on the geographic coordinates of a reference point and a target point, yielding the azimuth (in degrees) from the reference point to the target point.

```
findAzimuthGeogInput[refPtLat_, refPtLong_, otherPtLat_,
   otherPtLong_] := Module[{refPtVect, otherPtVect,
    northPole, vectorA, vectorB, θ1, θ2, vectorC, θ3, azimuth},
   refPtVect = {Cos[refPtLat Degree] Cos[refPtLong Degree],
     Cos[refPtLat Degree] Sin[refPtLong Degree], Sin[refPtLat Degree]};
   otherPtVect = {Cos[otherPtLat Degree] Cos[otherPtLong Degree],
     Cos[otherPtLat Degree] Sin[otherPtLong Degree],
     Sin[otherPtLat Degree]};
   northPole = {0, 0, 1};
   vectorA = Cross[northPole, refPtVect];
   vectorB = Cross[otherPtVect, refPtVect];
   θ1 = VectorAngle[vectorA, vectorB] (180 / π);
   θ2 = VectorAngle[vectorA, otherPtVect] (180 / π);
   vectorC = Cross[refPtVect, vectorA];
   θ3 = VectorAngle[vectorC, otherPtVect] (180 / π);
   azimuth = If[((θ2 == 90) ∨ (θ2 == 180)),
     If[(θ3 <= 90), 0, 180], If[(θ2 > 90), (360 - θ1), θ1]];
   azimuth ];
```

Exercise 2.4 (HW-04) What is the azimuth from Neenach in southern California (34°44'24"N, 118°37'24"W) in the direction of the shortest straight-line path to Pinnacles in west-central California (36°29'13"N, 121°10'01"W). ***The azimuth of a bearing is measured in a clockwise direction relative to true north.***

Write a *Mathematica* notebook that analyzes the input data to complete the following tasks.

(a) Convert the given geographic coordinates of Neenach and Pinnacles to decimal geographic coordinates (see section 2.2.3).

(b) Convert the decimal geographic coordinates to unit location vectors (**locVectNeenach** and **locVectPinnacles**), recalling that south latitudes and west longitudes are negative numbers (see section 2.2.3).

(c) Determine the circumferential distance of the shortest great-circle path from Neenach to Pinnacles (see section 2.3.4).

(d) Determine the azimuth of the shortest great-circle path from Neenach to Pinnacles (see section 2.4.3).

References

Boyd, S., and Vandenberghe, L., 2018, Introduction to Applied Linear Algebra — Vectors, Matrices, and Least Squares: Cambridge, UK, Cambridge University Press, 463 p.

Cazenave, A., 1995, Geoid, Topography, and Distribution of Landforms, *in* Ahrens, T.J. [editor], Global Earth Physics — A Handbook of Physical Constants: Washington, D.C., American Geophysical Union, AGU Reference Shelf 1, p. 32-39.

Davis, H.F., and Snider, A.D., 1987, Introduction to Vector Analysis [5th edition]: Boston, Massachusetts, Allyn and Bacon, 365 p.

Matthews, V., III, 1973, Pinnacles-Neenach correlation -- A restriction for models of the origin of the transverse ranges and the big bend in the San Andreas fault: Geological Society of America Bulletin, v. 84, p. 683-688.

Yoder, C.F., 1995, Astrometric and Geodetic Properties of Earth and the Solar System, *in* Ahrens, T.J. [editor], Global Earth Physics — A Handbook of Physical Constants: Washington, D.C., American Geophysical Union, AGU Reference Shelf 1, p. 1-31.